
Garbage Collection

The theme of this part of the course is “implementing Scheme on real computers.”  So far, we have been using a computer with 7 registers, a stack, a smattering of built-in operations, and a controller.  Most of our built-in operations are simple and can easily be built on real computers with only a small amount of work, and this makes our overall computer reasonably similar to modern computers.  But modern computers differ from our model computer in two ways:

a) Computers in the 1990s don’t have a special purpose stack; they have a stack register and put the actual contents of the stack into memory

b) CONS, CAR, and CDR are easy to implement if we have an infinite amount of memory, but most modern computers don’t even come close; so we have to worry about reusing memory that isn’t going to be needed any more.

We can buy computer memory in relatively large amounts.  It is divided into "addressable chunks" which can be individually read and written.  For our purposes, we will assume that the chunk is wide enough to hold a Scheme number or the address of another chunk of memory.  This isn't the way it really works, but it's a good approximation.  Into this memory we put the program we want to run (either the explicit control evaluator or the output of the compiler), the stack, and a “heap” that can be used to create pairs when we need them.  One way to arrange this in the memory is:

We build a stack by using a new register, called STACK, which has the address of an area of memory that we plan to use for this purpose. When we SAVE items on the stack we subtract from this address and use that as the new address for the stack.

What does CONS do?  It has to find two new spots for the CAR and the CDR.  We could use the stack, but that would interfere with our use of the stack to save and restore registers, since the usage pattern isn't the same.  Instead, we add yet one more register, FREE, and we split up the memory of the computer into three areas.
In order to do garbage collection (as well as to implement primitives like pair? and number?) we'll need to tell, somehow, if the contents of a memory location is a number or a CONS cell or something else.  A very simple way to do this (although not necessarily the most efficient way) is to use “typed data.”  Simply use part of every memory location (on the stack and the heap) to store the data type that is stored at that location, and use the rest of the location to store either a data value or the address where the value is found:

Some of the types we use are number (with the value stored in the value part), boolean (also with the value, either 0 or 1, in the value part), and cons (with the address of the CAR in the value, and the CDR stored at the next higher address).
We’re now ready to examine garbage collection.  To start, we have to have a good definition of "garbage".  We can use one based on our observation of the register machine itself: the state of the register machine, we've said, is the contents of the registers plus the stack.  We've built into our system the fact that saving these is enough to save the entire state of the system.  So that means we can start with these registers, copy everything they point to, recursively, and that's everything needed to continue computing.

So what is "garbage?"  It's whatever we do not find.  This is the fundamental idea behind garbage collection: you don't find the garbage!  Instead, we make a copy of the good stuff, and just let the part of memory that hold garbage be reused later.  We keep the good stuff instead of removing the bad stuff.

Let's assume that we somehow discover we're running low on space in the heap.  We grab the registers and make a LIST out of their contents. How would we got about copying this?

 (define (copy-list-1 l)


  (cond ((null? l) '())


        ((pair? l)


         (cons (copy-list-1 (car l))


               (copy-list-1 (cdr l))))


        (else l)))


(define x '(a b c))


(define y (cons x x))


(eq? (car y) (cdr y))  ; #T

(define z (copy-list-1 y))

(equal? y z)           ; #T

(eq? (car z) (cdr z))  ; #F

This works in this simple case, but it doesn't work because it changes the sharing relationships.  It also doesn't work on circular data structures.  Finally, it's recursive so it uses up stack space that may be limited.  Here’s another version in Scheme:

(define (copy-list-2 l)


  (let ((tag (list 'BEEN-HERE-DONE-THAT)))


    (define (copy l)


      (cond ((null? l) '())


            ((and (pair? l) (not (eq? (car l) tag)))


             (let ((the-car (car l))

                   (the-cdr (cdr l))

                   (result (cons 0 0)))

               (set-car! l tag)

               (set-cdr! l result)

               (set-car! result (copy the-car))

               (set-cdr! result (copy the-cdr))

               result))

            ((pair? l) (cdr l))

            (else l)))

    (copy l)))

(define x '(a b c))

(define y (cons x x))

y                      ; ((a b c) a b c)

(eq? (car y) (cdr y))  ; #T

(define z (copy-list-2 y))

z                      ; ((a b c) a b c)

(eq? (car z) (cdr z))  ; #T

(define x '(a b c))

(set-cdr! x x)

(define y (copy-list-2 x))

(eq? y (cdr y))

This version works.  Unfortunately, it's recursive.  At this point, instead of continuing to write the program in Scheme (in real use, of course, it’s written in assembly language), we’ll just describe the algorithm.  Use the attached sheet to try it out on your own.

Idea: we need to copy all of the “good stuff” into a new area.  So let's divide the heap into two parts, one in use until it fills up (“old area”), then copy the good stuff over into the other (“new area”), and then just continue running, using the new area until that fills up, and so forth.  How do we do the copy?  We'll use two new registers, FREE and SCAN.  We'll have three invariants:

Everything in the new part of memory with an address below SCAN is reachable from the computer’s registers and is perfectly ready for use.

1) Everything with an address at or above FREE is not in use.

2) Everything between (and including) SCAN and (but not including) FREE is reachable from the computer’s registers, and is a perfect copy of what it was before it was moved.

We start by copying all of the regular registers into the bottom of the new area of memory, pointing SCAN at the first location in the new area, and putting FREE just above the stuff just copied. Note that the invariants are true now (because there is nothing below SCAN).

When SCAN equals FREE, we are done, because everything at and above SCAN is free, everything below SCAN has been updated to correspond to its new location in memory.

Each step of the algorithm moves SCAN forward by one address.  In order to do this, we have to take the current thing that SCAN points to and make it ready for use in the new area.  There are three cases:

a) We are scanning a number or boolean (i.e. not a pointer to anything else), so just move the SCAN pointer forward.

We are scanning a pointer (it points to the old space).  We look there and see if it has a forwarding pointer:

if so, then it's already been moved so we just need to change the scanned item to point to where it now lives.

1) if not, then we have to move it.  Copy the thing it points to into wherever FREE points (and move FREE forward).  Replace the original value that SCAN pointed to with a forwarding pointer.

Can you argue that each of these cases maintains all three invariants?

You now have everything you need to make Scheme, or Java, run on a simple computer.  All we are assuming is that there are about 7 registers, some simple functions, and a large memory.  We put the stack into the memory, as well as the heap.  We use the stack to store registers temporarily, and we use the heap for everything else: CONS cells (including the environment), PROCEDURES, strings, very large numbers (bignums), arrays in Java.  The stack grows and shrinks as we save and restore registers.  The heap grows as we create new objects and shrinks when we garbage collect because it ran out of space to grow.

